
Tas et files de priorités

OPTION INFORMATIQUE - TP no 3.2 - Olivier Reynet

À la fin de ce chapitre, je sais :

R définir un tas-min et un tas-max

R trier un tableau pas tas

R utiliser un tas pour créer une file de priorité

R appliquer les tas à l’algorithme de Dijsktra

R programmer impérativement en OCaml

A Implémentation d’un tas max et tri par tas

A1. Créer une variable heap_test de type Array contenant les entiers de 1 à 10 dans l’ordre décroissant.
On génèrera automatiquement cette variable (pas in extenso). Est-ce un tas? Si oui, de quel type ?

Solution :

let heap_test = Array.init 10 (fun i -> 10 - i);;
val heap_test : int array = [|10; 9; 8; 7; 6; 5; 4; 3; 2; 1|]

C’est un tas-max.

A2. Écrire une fonction de signature swap : 'a array -> int -> int -> unit qui échange la place
de deux éléments dans un type Array. Tester cette fonction sur le tableau heap_test.

Solution :

let swap t i j = let tmp = t.(i) in t.(i) <- t.(j); t.(j) <- tmp;;

Comme en OCaml les Array sont indexés à partir de 0, on choisit de placer la racine du tas en 0. Puis,
si le père est k alors les fils seront en 2k +1 et 2k +2. Si un fils est en k, alors le père est en bk−1

2 c.

Il faut noter qu’un tas peut ne pas être rempli. Dans ce cas, on prendra soin de relever l’indice de la
première case non remplie.

A3. Écrire une fonction de signature up : 'a array -> int -> unit qui, si cela est possible, fait mon-
ter un élément d’après son indice dans le tas tout en préservant la structure du tas.

Solution :

let up heap k =
let rec aux i = match i with

| 0 -> ()
| _ -> let f = (i - 1)/2 in

if heap.(i) > heap.(f) then (swap heap i f ; aux f)
in aux k;;

A4. Écrire une fonction de signature down : 'a array -> int -> int -> unit qui, si cela est pos-
sible, fait descendre un élément dans un tas max tout en préservant la structure du tas. Le premier
entier de la signature est l’indice de la première case non remplie du tas, le second l’indice de l’élé-
ment à faire descendre.

Solution :

let down heap first_not_used k =
let rec aux i = match i with

| n when 2*n + 1 >= first_not_used -> ()
(* Out of bounds, it is done *)

| n when 2*n + 1 = (first_not_used - 1) -> if heap.(n) < heap.(2*n +
1) then swap heap (2*n + 1) n

(* just one last child *)
| _ -> begin (* two children *)

let f = if heap.(2*i + 1) > heap.(2*i + 2) then 2*i + 1
else 2*i + 2 in

if heap.(i) < heap.(f) then (swap heap i f; aux f;)
end

in aux k;;

On cherche à créer un tas de deux manières différentes :

• soit en considérant que le premier élément du tableau est un tas que l’on fait grossir avec les
éléments de droite du tableau,

• soit en considérant que les feuilles sont des tas à un élément que l’on fait grossir au fur avec
les éléments de gauche du tableau. On remarque dans ce cas que quelle que soit la taille n du
tas, l’élément d’indice n/2 est toujours la première feuille. On va donc insérer les n/2 premiers
éléments dans leurs feuilles et faire apparaître la structure de tas.

A5. Effectuer à la main la transformation du tableau [|0; 1; 2; 3; 4; 5; 6; 7; 8; 9|] en tas-max
par les deux méthodes.

A6. Écrire une fonction de signature heap_make_up : 'a array -> unit qui transforme un tableau en
un tas en faisant monter les éléments. Quelle la complexité de cette fonction?

Solution :

let heap_make_up t = for k = 1 to Array.length t - 1 do up t k done;;

La complexité au pire de cette fonction est linéaire en O(n logn) si n est la taille du tas. En effet,
dans le pire des cas, on fait remonter les n nœuds jusqu’à la racine, c’est-à-dire toute la hauteur
de l’arbre qui vaut logn.

A7. Écrire une fonction de signature heap_make_down : 'a array -> unit qui transforme un tableau
en un tas en faisant descendre les éléments. Quelle est la complexité de cette fonction 1 ?

Solution :

let heap_make_down t = let size = Array.length t in
for k = size/2 - 1 downto 0 do down t size k done;;

En considérant que le nombre de nœuds ayant la hauteur h est majoré par d n
2h+1 e ≤ 1+ nh

2h+1 , on
montre que :

Comme la série
∑
h

h
2h+1 converge et que p = blognc, on peut écrire que :

C (n) =O((logn)2 +n) =O(n) (2)

La complexité de la création d’un tas est linéaire si on fait descendre.

A8. Écrire une fonction de signature heap_sort : 'a array -> unit qui effectue le tri par tas d’un
tableau. Quelle la complexité de cette fonction?

Solution : La complexité du tri par tas vaut : C (n) = O(n +n log2 n) = O(n log2 n). On compte la
création du tas (linéaire) et la descente (log2 n) des n éléments dans le tas.

let heap_sort t =
heap_make_down t;
let size = Array.length t in

for k = size - 1 downto 0 do (swap t 0 k; down t k 0) done;;

A9. Tester le tri par tas sur un tableau de 10000 valeurs entières choisies aléatoirement entre 0 et 1000000.

Solution :

Random.init 42;;
let test = Array.init 100000 (fun i -> Random.full_int 1000000);;
heap_sort test;;
test;;

1. On suppose que le nombre de nœuds ayant la hauteur h est majoré par d n
2h+1 e

3 / 5



OPTION INFORMATIQUE TP no 3.2

B Implémentation d’une file de priorités

Pour implémenter une file de priorités à partir d’un tas, il est nécessaire de définir le type de données
que contient la file, puis de programmer les opérations de base de la file de priorités. Les opérations sur
une file de priorités sont :

1. créer une file vide,

2. insérer dans la file une valeur associée à une priorité (ENFILER),

3. sortir de la file la valeur associée la priorité maximale (ou minimale) (DÉFILER).

On considère une file de priorités dont les éléments sont des couples (v, p) et p est un entier naturel.
Plus p est faible, plus la priorité est élevée.

Par ailleurs, on se dote des types et des exceptions suivantes :

type 'a qdata = {value: 'a; priority: int};;
type 'a priority_queue = {mutable first_free: int; heap: 'a qdata array};;
exception EMPTY_PRIORITY_QUEUE;;
exception FULL_PRIORITY_QUEUE;;

B1. De quel type est le tas nécessaire à la construction de cette file de priorités?

Solution : Il s’agit d’un tas-min.

B2. Écrire une fonction de signature up : 'a array -> int -> unit qui, si cela est possible, fait mon-
ter un élément d’après son indice dans le tas tout en préservant la structure du tas. On veillera à bien
faire monter en fonction de la priorité.

Solution :

let up heap k =
let rec aux i = match i with

| 0 -> ()
| _ -> let p = (i - 1)/2 in

if heap.(i).priority < heap.(p).priority
then (swap heap i p ; aux p)

in aux k;;

B3. Écrire une fonction de signature down : 'a array -> int -> int -> unit qui, si cela est pos-
sible, fait descendre un élément dans un tas max tout en préservant la structure du tas. Le premier
entier de la signature est l’indice de la première case non remplie du tas, le second l’indice de l’élé-
ment à faire descendre. On veillera à bien faire descendre en fonction de la priorité.

Solution :

let down heap first_not_used k =
let rec aux i = match i with
| i when 2*i + 1 > first_not_used -> () (* Leave *)
| i when 2*i + 2 = first_not_used -> let f = 2*i + 1 in

if heap.(i).priority > heap.(f).priority then (swap heap i f;
aux f;)

| i -> begin
let f = if heap.(2*i + 1).priority < heap.(2*i + 2).priority

then 2*i + 1 else 2*i + 2 in
if heap.(i).priority > heap.(f).priority
then (swap heap i f; aux f;)

end
in aux k;;

B4. Écrire une fonction de signature make_priority_queue : int -> 'a * int -> 'a priority_queue
qui crée une file de priorité à n éléments initialisés à l’aide d’un type qdata donné en paramètre et

dont le premier élément libre est le premier du tas.

Solution :

let make_priority_queue n (v,p) = {
first_free = 0;
heap = Array.init n (fun i -> {value = v; priority=p})

};;

B5. Écrire une fonction de signature insert : 'a priority_queue -> 'a * int -> unit qui enfile
un élément dans la file de priorités. On veillera à préserver la structure du tas et à renvoyer une
exception si l’opération n’est pas possible.

Solution :

let insert pq (v,p) =
let size = Array.length pq.heap in
if pq.first_free + 1 > size then raise FULL_PRIORITY_QUEUE;
pq.heap.(pq.first_free) <- {value=v; priority=p};
up pq.heap pq.first_free;
pq.first_free <- pq.first_free + 1;;

B6. Écrire une fonction de signature get_min : 'a priority_queue -> 'aqui renvoie l’élément value
de priorité maximale. On veillera à préserver la structure du tas et à renvoyer une exception si l’opé-
ration n’est pas possible.

Solution :

let get_min pq =
if pq.first_free = 0 then raise EMPTY_PRIORITY_QUEUE;
let first = pq.heap.(0).value in
pq.first_free <- pq.first_free - 1;
pq.heap.(0) <- pq.heap.(pq.first_free);
down pq.heap pq.first_free 0;
first;;

